



Bitte erstellen Sie eine Excelmappe Ihres Namens *VornameNachname.xls* aus drei Tabellen: *Bestellung*, *Rechnung* und *Matrizen*. Darin nehmen Sie bitte die Bestellung von Frau Hehmke auf.

Bestellung: Hier entwerfen Sie eine Eingabemaske, in die der User alle für die Bestellung wesentlichen Informationen eingeben kann: Der User soll dabei nur die grau hinterlegten Felder ausfüllen. Schützen sie das Blatt entsprechend (OHNE Passwort!!!). Die Info-Tabelle D1:H11 soll aus der Matrizen-Tabelle ausgelesen werden. Sie soll Ihnen zeigen, welche ID-Nummer die CD jeweils hat. In Zeile 13 *ID* kommen die CD-ID-Nummern, in Zeile 14 *Menge* wie viel davon je bestellt wurden.

Rechnung: Die Tabelle *Rechnung* wird als Rechnungs-Anschreiben ausgedruckt und am gleichen Tag auch abgeschickt. Deshalb soll sie den Anforderungen eines Geschäftsbriefs genügen. Ihr wichtigster Teil sieht so aus wie unten dargestellt. *In das Rechnungsformular werden keine Bestelldaten eingeben!* Diese Informationen werden aus der Tabelle *Bestellungen* ausgelesen. Anrede und Rechnungstext stehen in Spalte A (dabei können Sie natürlich Zellen *verbinden*) Fehlende Angaben ergänzen Sie bitte wirklichkeitsnah. Die Bestellungsdaten finden Sie in unter bei **Bestellung**.

Matrizen: Die Tabelle *Matrizen* besteht aus drei Matrizen: *CDs*, *Porto* und *Rabatte*.

Ihr CD-Sortiment:	
Anne Clark/ Anagramm	13,49 €
Bravo Hits 27	25,49 €
Bravo Hits 40	18,49 €
Deep Trance Vol. 4	25,49 €
Eminem/ 8 Mile	23,49 €
Eminem/ Lose Yourself	3,49 €
Eminem/ The Eminem Show	15,49 €
Fettes Brot/ Amnesie	17,49 €
Fluke/ Risotto	8,49 €
Oldfield, Mike/ Q.E.2	12,49 €

Porto: Eine kleine Warensendung bis 80 gr kostet 55Cent, eine große bis 249 gr kommt auf 1€. Ein kleines Packet bis 499 gr kostet 1,44€. Ein mittleres Packet bis 1kg kostet 2,20€, ein großes bis 5 kg kommt auf 6,70€. Eine kleine Sendung bis maximal 10 kg kostet 13€, eine große über 10 kg wird je nach Gewicht abgerechnet, - aber sie stellen dem Großkunden hier kein Porto mehr in Rechnung. Eine CD wiegt 75gr.

Rabatte:
 Um Großkunden zu locken, gewähren Sie ab einem Bestellwert von 500€ 5% Rabatt. Kleinere Großkunden bekommen ab 100€ 2% und ab 250€ 3%

Bestellung:
 Grete Hehmke, geborene Friese, bestellte am 15.3.2003 für ihren Kegelclub 3x BravoHits 40, 5x Anne Clark, 10x Eminem *Loose yourself* und 15x Fluke. Sie wohnt in 12345 Seniorenhausen in der Schaukelstuhlstraße 7.

Bestelldaten	Info
	1 Anne Clark/ Anagramm 13,49 €
	2 Bravo Hits 27 25,49 €
	3 Bravo Hits 40 18,49 €
	4 Deep Trance Vol. 4 25,49 €
	5 Eminem/ 8 Mile 23,49 €
	6 Eminem/ Lose Yourself 3,49 €
	7 Eminem/ The Eminem Sh 15,49 €
	8 Fettes Brot/ Amnesie 17,49 €
	9 Fluke/ Risotto 8,49 €
	10 Oldfield, Mike/ Q.E.2 12,49 €
Anrede	Frau Dr.
Name	Webbern
Vorname	Hanna
Straße und Hausnummer	Musikalienweg 18
PLZ und Ort	98765 Ohrenhausen
Geschlecht	W
Bestelldatum	19.03.2003
Produktnummern	1 2 3 4 5
ID	6 4 5 1
Menge	1 5 3 1 10

Rechnung

Artikel	Einzelpreis	Stück	Postenpreis
Eminem/ Lose Yourself	3,49 €	1	3,49 €
Deep Trance Vol. 4	25,49 €	5	127,45 €
Eminem/ 8 Mile	23,49 €	3	70,47 €
Anne Clark/ Anagramm	13,49 €	1	13,49 €
<i>Zwischensumme:</i>	- €	0	- €
		10	214,90 €
		Porto	2,20 €
		Rabatt (2%)	- 4,34 €
		Rechnungsbetrag	212,76 €

Sehr geehrte Frau Dr. Webbern,

bitte überweisen Sie uns in den nächsten Tagen 212,76 EUR auf unser unten angegebenes Konto.



Lösung:

Anlage der Tabelle Matrizen:

Eine SVerweis-Matrix muss in der ersten Spalte immer eine aufsteigende Folge von Zahlen stehen haben.

Deshalb müssen Sie die **CDs-Matrix** vom Arbeitsblatt um eine Zahlenspalte erweitern. Die Zahlen von 1-10 stehen anschließend für die ID-Nummer, die Sie später in der Tabelle *Bestellung* verwenden werden. Der Preis muss als Zahl im Währungsformat in die dritte Spalte geschrieben werden, weil sie später mit diesem Preis noch rechnen müssen. Fügen Sie eine neue Zeile ein, damit bei weniger als 5 Bestellungen kein Fehler auftritt: wenn nur vier unterschiedliche CDs bestellt werden, dann wird bei die fünfte Stelle keine Fehlermeldung enthalten, sondern anzeigen, was Sie unter der ID 0 in die zweite Spalte eingetragen haben.

PortoMatrix

Gewicht in Gramm	Porto in EUR
1	0,55
81	1,00
250	1,44
500	2,20
1001	6,70
5001	13,00
10001	0

Vorsicht bei der Erstellung der Porto-Matrix: Hier werden im Text *Bis-Werte* angegeben: eine SVerweis-Matrix verlangt aber *Ab-Werte*: Tragen Sie deshalb in die zweite Spalte alle zur Verfügung stehenden Porto-Preise ein und überlegen Sie dann, *ab* welchem Gewicht sie jeweils zugeordnet werden: Ab wie viel Gramm müssen 55 Cent Porto bezahlt werden? Ab welchem Gewicht müssen 100 Cent bezahlt werden ... Bitte wechseln Sie in einer Matrix nicht die Einheit, sondern bleiben Sie entweder bei Cent oder EUR, bei Gramm oder Kilogramm.

Die Rabatt-Matrix ist einfach abzuschreiben – hier sind bereits *Ab-Werte* gegeben.

► Sie erleichtern sich die Arbeit, wenn Sie ihren Matrizen einen Namen geben (*Einfügen/Namen/definieren*) (Möglichst nur die Wertezellen benennen! Z.B.: A3:C13)

Anlage der Tabelle Bestellung:

Bilden Sie das Modell nach – tragen Sie dabei gleich die Daten von Frau Hehmke ein.

Die CD-Matrix lesen Sie aus der *Matrizen-Tabelle* aus.

Man liest den Zellinhalt einer Zelle am einfachsten aus, indem man in die Zelle, in der das Ergebnis stehen soll (hier z.B. D2), ein Istgleich-Zeichen = ein und klickt dann auf die gewünschte Zelle. Am Ende muss man entweder die Enter-Taste (↵) betätigen oder das grüne Häkchen neben der Bearbeitungsleiste anklicken.

Einem Bezug auf eine Zelle in einer anderen Tabelle muss man den Namen dieser Tabelle gefolgt von einem Ausrufe-Zeichen voranstellen. In D2 also etwa: =Matrizen!A3

Produktnummern	1	2	3	4	5	
ID	3	1	6	9		Σ
Menge	3	5	10	15		33

Die Bestellung lautet mit den eingetragenen IDs: Die ID ist die der CD in der CD-Matrix jeweils zugeordnete Nummer.

Die CD *Bravo Hits 40* hat die Nummer 3, die *Anne Clark* CD die Nummer 1 ...

Ganz rechts unter dem Summenzeichen (*Einfügen/ Symbol/Schriftart Symbol ...*) wird die insgesamt bestellte CD-Menge aufsummiert: =SUMME(B14:F14)

Anlage der Tabelle Rechnung:

Stellen Sie die Seitenränder ein unter *Datei/ Seite einrichten/ Seitenränder*

Geben Sie als Absender Ihr Geschäft ein (keine genauen Vorgaben), lesen Sie die Adresse des Kunden aus der Tabelle *Bestellung* aus: Dazu tippen Sie in die Zelle, in der eine bestimmte Information stehen soll, ein *Istgleich-Zeichen* = ein und klicken dann auf die Zelle in der Tabelle *Bestellung*, in der die gewünschte Information eingetragen ist. In der Zelle A4 der Tabelle *Rechnung* sollten Vorname und Name der Kundin stehen. Diese Informationen haben Sie in die Zellen B6 und B7 der Tabelle *Bestellung* eingetragen. Diese Zellen lesen Sie mit der folgenden Formel aus: =Bestellung!B7 & " " & Bestellung!B6 Zum Kombinieren von Name und Vorname verwenden Sie das *kaufmännische UND* & (⇧) (6). Fügen Sie eine Leerstelle " " (Leerstelle zwischen Anführungszeichen) zwischen Vor- und Nachname ein.

Als Rechnungsdatum soll das je aktuelle Datum eingetragen werden: =Heute()

	A	B	C	D	E
18		Artikel	Einzelpreis	Stück	Postenpreis
19	=Bestellung!B13	=SVERWEIS(\$A19;CDMatrix;2)	=SVERWEIS(\$A19;CDMatrix;3)	=Bestellung!B14	=D19*C19
	Schriftfarbe weiß -> unsichtbar	Das \$-Zeichen vor dem Spalten-Buchstaben aber nicht vor der Zeilen-Zahl: So kann die Formel leicht in die Einzelpreis-Spalte gezogen werden, wo nur noch der Spaltenindex angepasst werden muss		Aus der Tabelle <i>Bestellung</i> auslesen	

Sie haben in der Tabelle *Bestellung* in Zelle B13 die ID-Nummer der ersten bestellten CD aus der CD-Matrix eingetragen. Übernehmen Sie diese ID-Nummer in die Zelle A19 der *Rechnungstabelle* – schreiben Sie diese in weißer Textfarbe, dann erscheint sie auf weißem Hintergrund unsichtbar. In der Zelle B19 soll nun der Titel der CD, deren Nummer in A19 steht, mit Hilfe der SVerweisnummer ausgegeben werden. Der SVerweis-Formel werden dabei drei mit Semikolon getrennte Argumente übergeben: zuerst die ID-Nummer in A19, dann die Matrix CDMatrix und schließlich die Spaltennummer dieser Matrix, in der der Wert steht, der ausgegeben werden soll, das ist hier die Spalte 2. In C19 verwenden Sie die gleiche

CDMatrix

ID	Titel	Einzelpreis in €
0	= " " "	0
1	Anne Clark/ Anagramm	13,49 €
2	Bravo Hits 27	25,49 €
3	Bravo Hits 40	18,49 €
4	Deep Trance Vol. 4	25,49 €
5	Eminem/ 8 Mile	23,49 €
6	Eminem/ Lose Yourself	3,49 €
7	Eminem/ The Eminem Show	15,49 €
8	Fettes Brot/ Amnesie	17,49 €
9	Fluke/ Risotto	8,49 €
10	Oldfield, Mike/ Q.E.2	12,49 €

RabattMatrix

Bestellwert in EUR	Rabatt
0	0
100	0,02
250	0,03
500	0,05
Formatieren Sie die Rabatt-Spalte als Prozent % ohne Komma	



Formel wie in B19 – nur der Spaltenindex ändert sich von 2 auf 3, denn in der dritten Spalte der CDMatrix steht der Einzelpreis der ausgewählten CD.

In B14 steht, wie oft die erste CD bestellt wurde.. In E multiplizieren Sie die bestellte Menge mit dem Einzelpreis.

Bitte verfahren Sie in den Zeilen bis 23 analog: Achtung: die D-Spalte/ Stück können Sie nicht einfach herunterziehen, weil die Menge in der Tabelle Bestellung nicht untereinander, sondern nebeneinander steht!

In E24 addieren Sie die Postenpreise zu einer Zwischensumme. In E24 zählen Sie die insgesamt bestellten CDs. Erledigen Sie beides komfortabel mit der Summen-Funktion =SUMME(D19:D23) bzw =SUMME(E19:E23)

Ihr Kunde soll auch die Portokosten übernehmen. Diese berechnen sich aus dem Gewicht der Bestellung. Weil jede Ihrer CDs 75 Gramm wiegt, müssen Sie die Menge der insgesamt bestellten CDs (in Zelle D24) mit 75 multiplizieren, dann erhalten Sie das Gewicht Ihrer Sendung. Dieses Gewicht ist das Suchkriterium, das bestimmt, welches Porto zu bezahlen ist. Die Porto-Matrix haben Sie vorhin PortoMatrix genannt, dort steht in der zweiten Spalte das Porto in EUR: Also ergibt sich als Porto in der Zelle E25: =SVerweis(D24*75;PortoMatrix;2)

In D26 soll hinter Rabatt in Klammern angezeigt werden, wie viel Rabatt Sie gewähren.

Das erfahren Sie mit Hilfe der SVerweis-Funktion: Das Suchkriterium ist der Bestellwert, der in der Zelle E24 steht. Die einschlägige Matrix, in der Bestellwerte und Rabatte verbunden sind, heißt RabattMatrix, dort steht der Rabatt in der zweiten Spalte. Der Rabatt der aktuellen Bestellung beträgt also =SVerweis(E24;RabattMatrix;2)

Damit dieser Wert in D26 mit angezeigt wird, müssen Sie ihn mit dem anderen Text kombinieren:

D26: ="Rabatt (" & SVerweis(E24;RabattMatrix;2) & ")"

Der Rabatt muss noch auf die Rechnungssumme bezogen werden, indem Sie ihn mit ihr multiplizieren. Weil der eingeräumte Rabatt der Rechnungssumme abgezogen wird, müssen Sie ihn mit (-1) multiplizieren. In E26 steht also:

= SVerweis(E24;RabattMatrix;2)*E24*(-1)

Der Rechnungsbetrag ergibt sich nun als Summe von Zwischensumme (E24), Porto (E25) und Rabatt (E26):

=Summe(E24:E26). Heben Sie diesen Rechnungsbetrag optisch hervor.

Zum Abschluss müssen Sie noch einen kleinen Text schreiben. Je nach Geschlecht müssen Sie die Anrede verändern:

Wenn Sie eine Kundin anschreiben, dann müssen Sie „Sehr geehrte“ & Anrede & Nachname kombinieren; wenn Sie aber einem Kunden schreiben, muss es „Sehr geehrter“ & Anrede & Nachname heißen – hier ist ein r hinzugekommen: Also: ="Sehr geehrte" & WENN(Bestellung!B10="m";"r ";" ") & Bestellung!B5 & " " & Bestellung!B6 & ", "

Wenn also in der Tabelle Bestellung bei Geschlecht (B10) ein m eingetragen wurde, dann handelt es sich um einen Mann und also müssen Sie Sehr geehrter schreiben, dem sehr geehrte also noch ein r anhängen (mit Leerstelle, damit die Anrede nicht ohne Zwischenraum an das Vorwort stößt). Ist in B10 aber ein w, dann soll nur eine Leerstelle angehängt werden.

Vergessen Sie nicht zwischen Anrede und Nachnamen eine Leerstelle einzufügen und am Ende ein Komma zu platzieren! Im Anschreiben soll die Rechnungssumme genannt werden. Das Problem dabei ist, dass diese Rechnungssumme unformatiert ausgelesen wird. Wenn Sie als letzte Nachkommastelle eine 0 besitzt, dann wird diese nicht angezeigt, und es wirkt unprofessionell, wenn in der Rechnung z.B. 235,4 € stünde: Währungen werden normalerweise mit zwei Nachkommastellen angezeigt.

In F27 soll der auf zwei Nachkommastellen gerundete Rechnungsbetrag mit hundert multipliziert werden. Die Zelle soll nicht als Währung formatiert sein: =RUNDEN(E27*100;0)

Sie müssen drei Fälle unterscheiden:

Fall 1: Der hundertfache Rechnungsbetrag in F27 besitzt am Ende zwei Nullen

Dann müssen dem auf zwei Nachkommastellen gerundete Betrag ein Komma und zwei Nullen angehängt werden

Wenn(rechts(F27;2)="00";Runden(E27;2)&"00")

Fall 2: Der hundertfache Rechnungsbetrag in F27 besitzt am Ende eine Null

Dann muss dem auf zwei Nachkommastellen gerundete Betrag einfach eine Null angehängt werden

Wenn(rechts(F27;1)="0";Runden(E27;2)&"0")

Fall 3: Der hundertfache Rechnungsbetrag in F27 besitzt am Ende keine Null

Dann kann der auf zwei Nachkommastellen gerundete Betrag einfach übernommen werden

Wenn(rechts(F27;1)<>"0";Runden(E27;2))

Bei diesen Wenn-Formeln fehlen noch die Sonst-Werte, die verschachtelt werden müssen: Schreiben Sie in E28:

=WENN(RECHTS(F27;2)="00";RUNDEN(E27;2)&"00"; WENN(RECHTS(F27;1)="0";RUNDEN(E27;2)&"0";RUNDEN(E27;2))) & " €"

(Diese Formel ist natürlich etwas für sehr gute Spezialisten!)

Setzen Sie die Schriftgröße von E28 und F27 auf weiß um sie unsichtbar zu machen.

In A32 steht also: ="bitte überweisen Sie uns in den nächsten Tagen " & F28 & " auf unser unten angegebenes Konto."

Vergessen Sie nicht Ihre Bankverbindung am Ende der Rechnung zu nennen!

Die SVerweis-Funktion basiert auf einer **Zuordnungs-Tabelle** (die man Matrix nennt): darin werden *aufsteigenden* Zahlen-Werten in einer ersten Spalte andere Werte zugeordnet: z.B. einem Gewicht (Spalte 1) soll ein Porto (Spalte 2) zugeordnet werden oder Der Höhe der Geschwindigkeitsübertretung (Spalte 1) soll eine Strafe (Spalte 2) zugeordnet werden oder Einer Produktnummer (Spalte 1) soll ein Produkt (Spalte 2), dessen Preis (Spalte 3) und dessen Gewicht (Spalte 4) zugeordnet werden.

Die SVerweis –Funktion erwartet, dass Sie ihr **drei Argumente** in genau der folgenden Reihenfolge mit einem Semikolon getrennt übergeben: **zuerst** das Suchkriterium, das ist der Wert der ersten Spalte, der es ermöglicht, den diesem Wert zugeordneten anderen Wert aus den anderen Spalten zurückzugeben. Damit findet Excel die Zeile, aus der es den Wert auslesen soll. Diese Zeile befindet sich in der Matrix, die als **zweites Argument** übergeben wird. Das ist die Zuordnungs-Tabelle, in der den Werten der ersten Spalte andere Werte in den anderen Spalten zugeordnet werden. **Schließlich** der Spaltenindex, der die Spalte bestimmt, aus der der Rückgabewert entnommen werden soll (meist 2 für die zweite Spalte).